```
(*Video 1: Concave up/down *)
```

In[185]:=
```
f[x_] := 30 + 6 x - 15 x^2 + 3 x^3 + x^4;
Manipulate[Grid[{{Row[{"Value of second derivative is f''[x0]=", f''[x0]}]},
   {Plot[{f[x], f'[x0] (x - x0) + f[x0]}, {x, -10, 5},
     PlotStyle → {Normal, Gray}, PlotRange → {-200, 200}, ImageSize → 500]}},
  Spacings → {1, 1}, Frame → All], {{x0, -3}, -10, 5}]
```

Out[186]=




```
(* f''>0 / Tangent line locally below graph  --- function is concave up*)
(* f''<0 / Tangent line locally above graph  --- function is concave down*)
(* Points x0 where concavity changes from
 up to down or down to up are INFLECTION POINTS
*)
```

In[174]:= `Plot[{f[x], f'[x], f''[x]}, {x, -10, 5}]`

Out[174]=



In[191]:= `f''[x]`

Out[191]= $-30 + 18x + 12x^2$

In[192]:= `Factor[f''[x]]`

Out[192]= $6(-1 + x)(5 + 2x)$

In[194]:=
```
Reduce[f''[x] > 0](*With Mathematica we can use Reduce*)
Reduce[f''[x] < 0]
```

Out[194]= $x < -\frac{5}{2} \,||\, x > 1$

Out[195]= $-\frac{5}{2} < x < 1$

```
(*Inflection points:*)
Solve[f''[x] == 0, x]
```

Out[196]= $\left\{\left\{x \to -\frac{5}{2}\right\}, \{x \to 1\}\right\}$

In[197]:=
```
concaveup = Show[
  Plot[f[x], {x, -10, -5 / 2}, PlotRange → {{-10, 5}, {-200, 200}}, PlotStyle → Blue],
  Plot[f[x], {x, 1, 5}, PlotRange → {{-10, 5}, {-200, 200}}, PlotStyle → Blue]]
```

Out[197]=



In[171]:=
```
concavedown =
 Plot[f[x], {x, -5 / 2, 1}, PlotRange → {{-5, 5}, {-200, 200}}, PlotStyle → Red]
```

Out[171]=



In[237]:=
```
Show[concaveup, concavedown]
```

Out[237]=



```
(*Video 2: Second derivative test*)

(* If x=x0 is a critical point of f[x], that is f'[x0]==0, and,
 f''[x0]>0 then x=x0 is a local minimum,
 f''[x0]<0 then x=x0 is a local maximum.
   However: if f''[x0]==0, then test is inconclusive (anything can happen!)
*)

(*Find the critical points and classify them into local min,
local max, or neither:*)
f[x_] := 3 x^5 - 5 x^3 + 3
```

In[211]:= `Plot[f[x], {x, -2, 2}]`

Out[211]=



In[216]:= `f'[x]`

Out[216]= $-15 x^2 + 15 x^4$

In[215]:= `Solve[f'[x] == 0, x]`

Out[215]= `{{x → -1}, {x → 0}, {x → 0}, {x → 1}}`

```
(*There are 3 critical points: x=-1, x=0, and x=1 *)
```

In[219]:= `f''[x] /. {x → -1}`

Out[219]= -30

```
(* x=-1 is a local maximum!*)
```

In[220]:=
```
f''[x] /. {x → 1}
```
Out[220]= 30

```
(* x=1 is a local minimum!*)
```

In[221]:=
```
f''[x] /. {x → 0}
```
Out[221]= 0

```
(*test is inconclusive... *)
```

In[222]:=
```
Plot[f[x], {x, -1 / 2, 1 / 2}]
```
Out[222]=



In[228]:=
```
Manipulate[Grid[{{Row[{"Value of second derivative is f''[x0]=", f''[x0]}]},
   {Plot[{f[x], f'[x0] (x - x0) + f[x0]}, {x, -2, 2},
     PlotStyle → {Normal, Gray}, PlotRange → {-5, 5}, ImageSize → 500]}},
  Spacings → {1, 1}, Frame → All], {{x0, 0}, -2, 2}]
```

Out[228]=



In[229]:=
```
Plot[f'[x], {x, -1 / 2, 1 / 2}]
```

Out[229]=




```
(* Tangent line has negative slope both before and after x=0,
so x=0 is not a local min, nor a local max. *)
```

In[230]:=
```
f''[x]
```
Out[230]= $-30\,x + 60\,x^3$

In[241]:=
```
Plot[{f[x], f'[x], f''[x]}, {x, -3 / 2, 3 / 2}]
```
Out[241]=


In[234]:=
```
Solve[f''[x] == 0, x]
```
Out[234]= $\left\{\{x \to 0\}, \left\{x \to -\frac{1}{\sqrt{2}}\right\}, \left\{x \to \frac{1}{\sqrt{2}}\right\}\right\}$

```
(*All of these 3 points are inflection points*)

Reduce[f''[x] > 0, x] (*concave up!*)
```
Out[235]= $-\frac{1}{\sqrt{2}} < x < 0 \,||\, x > \frac{1}{\sqrt{2}}$

```
Reduce[f''[x] < 0, x] (*concave down!*)
```
Out[236]= $x < -\frac{1}{\sqrt{2}} \,||\, 0 < x < \frac{1}{\sqrt{2}}$

In[242]:=
```
concaveup = Show[
  Plot[f[x], {x, -1/√2, 0}, PlotRange → {{-3 / 2, 3 / 2}, {-5, 5}}, PlotStyle → Blue],
  Plot[f[x], {x, 1/√2, 3 / 2}, PlotRange → {{-3 / 2, 3 / 2}, {-5, 5}}, PlotStyle → Blue]]
```

Out[242]=



In[243]:=
```
concavedown = Show[Plot[f[x], {x, -3 / 2, -1/√2},
   PlotRange → {{-3 / 2, 3 / 2}, {-5, 5}}, PlotStyle → Red],
  Plot[f[x], {x, 0, 1/√2}, PlotRange → {{-3 / 2, 3 / 2}, {-5, 5}}, PlotStyle → Red]]
```

Out[243]=



In[244]:= `Show[concavedown, concaveup]`

Out[244]=